Rules for integrands of the form $(d\,x)^m \left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^p$

1: $\int x^m\,(a\,x^n + b\,x^n + c\,x^n)^p\,dx$

- **Rule:**

$$\int x^m\,(a\,x^n + b\,x^n + c\,x^n)^p\,dx \longrightarrow \int x^m\,((a+b+c)\,x^n)^p\,dx$$

- **Program code:**

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_.,x_Symbol] :=
  Int[x^m*((a+b+c)*x^n)^p,x] /;
FreeQ[{a,b,c,m,n,p},x] && EqQ[q,n] && EqQ[r,n]
```

2: $\int x^m\left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^p\,dx$ when $p \in \mathbb{Z}$

- **Rule: If $p \in \mathbb{Z}$, then**

$$\int x^m\left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^p\,dx \longrightarrow \int x^{m+p\,q}\left(a + b\,x^{n-q} + c\,x^{2\,(n-q)}\right)^p\,dx$$

- **Program code:**

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_.,x_Symbol] :=
  Int[x^(m+p*q)*(a+b*x^(n-q)+c*x^(2*(n-q)))^p,x] /;
FreeQ[{a,b,c,m,n,q},x] && EqQ[r,2*n-q] && IntegerQ[p] && PosQ[n-q]
```

3. $\int \frac{x^m}{\sqrt{a x^q + b x^n + c x^{2n-q}}} \, dx$ when $q < n \wedge b^2 - 4 a c \neq 0$

1: $\int \frac{x^m}{\sqrt{a x^q + b x^n + c x^{2n-q}}} \, dx$ when $q < n \bigwedge b^2 - 4 a c \neq 0 \bigwedge m == \frac{q}{2} - 1$

- **Derivation: Integration by substitution**
- **Basis: If** $m == \frac{q}{2} - 1$**, then** $\frac{x^m}{\sqrt{a x^q+b x^n+c x^{2n-q}}} == -\frac{2}{n-q} \text{Subst}\left[\frac{1}{4a-x^2}, x, \frac{x^{m+1} (2a+b x^{n-q})}{\sqrt{a x^q+b x^n+c x^{2n-q}}}\right] \partial_x \frac{x^{m+1} (2a+b x^{n-q})}{\sqrt{a x^q+b x^n+c x^{2n-q}}}$
- **Rule: If** $q < n \bigwedge b^2 - 4 a c \neq 0 \bigwedge m == \frac{q}{2} - 1$**, then**

$$\int \frac{x^m}{\sqrt{a x^q + b x^n + c x^{2n-q}}} \, dx \longrightarrow -\frac{2}{n-q} \text{Subst}\left[\int \frac{1}{4a - x^2} \, dx, x, \frac{x^{m+1} (2a + b x^{n-q})}{\sqrt{a x^q + b x^n + c x^{2n-q}}}\right]$$

Program code:

```
Int[x_^m_./Sqrt[a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.],x_Symbol] :=
  -2/(n-q)*Subst[Int[1/(4*a-x^2),x],x,x^(m+1)*(2*a+b*x^(n-q))/Sqrt[a*x^q+b*x^n+c*x^r]] /;
FreeQ[{a,b,c,m,n,q,r},x] && EqQ[r,2*n-q] && PosQ[n-q] && NeQ[b^2-4*a*c,0] && EqQ[m,q/2-1]
```

2: $\int \frac{x^m}{\sqrt{a x^q + b x^n + c x^{2n-q}}} \, dx$ when $q < n$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{x^{q/2} \sqrt{a+b x^{n-q}+c x^{2(n-q)}}}{\sqrt{a x^q+b x^n+c x^{2n-q}}} == 0$

Rule: If $q < n$**, then**

$$\int \frac{x^m}{\sqrt{a x^q + b x^n + c x^{2n-q}}} \, dx \longrightarrow \frac{x^{q/2} \sqrt{a + b x^{n-q} + c x^{2(n-q)}}}{\sqrt{a x^q + b x^n + c x^{2n-q}}} \int \frac{x^{m-q/2}}{\sqrt{a + b x^{n-q} + c x^{2(n-q)}}} \, dx$$

Program code:

```
Int[x_^m_./Sqrt[a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.],x_Symbol] :=
  x^(q/2)*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))]/Sqrt[a*x^q+b*x^n+c*x^(2*n-q)]*
    Int[x^(m-q/2)/Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))],x] /;
FreeQ[{a,b,c,m,n,q},x] && EqQ[r,2*n-q] && PosQ[n-q] && (EqQ[m,1] && EqQ[n,3] && EqQ[q,2]  ||
  (EqQ[m+1/2] || EqQ[m,3/2] || EqQ[m,1/2] || EqQ[m,5/2]) && EqQ[n,3] && EqQ[q,1])
```

4: $\int \frac{x^{\frac{3(n-1)}{2}}}{\left(a x^{n-1}+b x^n+c x^{n+1}\right)^{3/2}} \, dx$ when $b^2-4 a c \neq 0$

- **Rule: If $b^2-4 a c \neq 0$, then**

$$\int \frac{x^{\frac{3(n-1)}{2}}}{\left(a x^{n-1}+b x^n+c x^{n+1}\right)^{3/2}} \, dx \longrightarrow -\frac{2 x^{\frac{n-1}{2}} (b+2 c x)}{\left(b^2-4 a c\right) \sqrt{a x^{n-1}+b x^n+c x^{n+1}}}$$

- **Program code:**

```
Int[x_^m_./(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^(3/2),x_Symbol] :=
  -2*x^((n-1)/2)*(b+2*c*x)/((b^2-4*a*c)*Sqrt[a*x^(n-1)+b*x^n+c*x^(n+1)]) /;
FreeQ[{a,b,c,n},x] && EqQ[m,3*(n-1)/2] && EqQ[q,n-1] && EqQ[r,n+1] && NeQ[b^2-4*a*c,0]
```

5: $\int \frac{x^{\frac{3n-1}{2}}}{\left(a x^{n-1}+b x^n+c x^{n+1}\right)^{3/2}} \, dx$ when $b^2-4 a c \neq 0$

- **Rule: If $b^2-4 a c \neq 0$, then**

$$\int \frac{x^{\frac{3n-1}{2}}}{\left(a x^{n-1}+b x^n+c x^{n+1}\right)^{3/2}} \, dx \longrightarrow \frac{x^{\frac{n-1}{2}} (4 a+2 b x)}{\left(b^2-4 a c\right) \sqrt{a x^{n-1}+b x^n+c x^{n+1}}}$$

- **Program code:**

```
Int[x_^m_./(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^(3/2),x_Symbol] :=
  x^((n-1)/2)*(4*a+2*b*x)/((b^2-4*a*c)*Sqrt[a*x^(n-1)+b*x^n+c*x^(n+1)]) /;
FreeQ[{a,b,c,n},x] && EqQ[m,(3*n-1)/2] && EqQ[q,n-1] && EqQ[r,n+1] && NeQ[b^2-4*a*c,0]
```

6: $\int x^m \left(a x^{n-1} + b x^n + c x^{n+1}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+ \land m + p (n - 1) - 1 == 0$

Derivation: Generalized trinomial recurrence 3a with $A = 0$, $B = 1$, $q = n - 1$ and $m + p (n - 1) - 1 == 0$

Rule: If $q < n \land p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+ \land m + p (n - 1) == 1$, then

$$\int x^m \left(a x^{n-1} + b x^n + c x^{n+1}\right)^p dx \longrightarrow \frac{x^{m-n} \left(a x^{n-1} + b x^n + c x^{n+1}\right)^{p+1}}{2 c (p + 1)} - \frac{b}{2 c} \int x^{m-1} \left(a x^{n-1} + b x^n + c x^{n+1}\right)^p dx$$

Program code:

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  x^(m-n)*(a*x^(n-1)+b*x^n+c*x^(n+1))^(p+1)/(2*c*(p+1)) -
  b/(2*c)*Int[x^(m-1)*(a*x^(n-1)+b*x^n+c*x^(n+1))^p,x] /;
FreeQ[{a,b,c},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] &&
  RationalQ[m,p,q] && EqQ[m+p*(n-1)-1,0]
```

7. $\int x^m \left(a x^q + b x^n + c x^{2 n-q}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+ \land p > 0$

1: $\int x^m \left(a x^q + b x^n + c x^{2 n-q}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+ \land p > 0 \land m + p q + 1 == n - q$

Derivation: Generalized trinomial recurrence 1b with $A = 0$, $B = 1$ and $m + p q + 1 == 0$

Rule: If $q < n \land p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+ \land p > 0 \land m + p q + 1 == n - q$, then

$$\int x^m \left(a x^q + b x^n + c x^{2 n-q}\right)^p dx \longrightarrow$$

$$\frac{x^{m-n+q+1} \left(b + 2 c x^{n-q}\right) \left(a x^q + b x^n + c x^{2 n-q}\right)^p}{2 c (n - q) (2 p + 1)} - \frac{p \left(b^2 - 4 a c\right)}{2 c (2 p + 1)} \int x^{m+q} \left(a x^q + b x^n + c x^{2 n-q}\right)^{p-1} dx$$

Program code:

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  x^(m-n+q+1)*(b+2*c*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^p/(2*c*(n-q)*(2*p+1)) -
  p*(b^2-4*a*c)/(2*c*(2*p+1))*Int[x^(m+q)*(a*x^q+b*x^n+c*x^(2*n-q))^(p-1),x] /;
FreeQ[{a,b,c},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GtQ[p,0] &&
  RationalQ[m,q] && EqQ[m+p*q+1,n-q]
```

2: $\int x^m \left(a x^q + b x^n + c x^{2 n-q}\right)^p dx$ when

$q<n \wedge p\notin\mathbb{Z} \wedge b^2-4ac\neq 0 \wedge n\in\mathbb{Z}^+ \wedge p>0 \wedge m+pq+1>n-q \wedge m+p(2n-q)+1\neq 0 \wedge m+pq+(n-q)(2p-1)+1\neq 0$

Derivation: Generalized trinomial recurrence 1b with $A=0$, $B=1$ and $m=m-n+q$

Rule: If $q<n \wedge p\notin\mathbb{Z} \wedge b^2-4ac\neq 0 \wedge n\in\mathbb{Z}^+ \wedge p>0 \wedge m+pq+1>n-q \wedge m+p(2n-q)+1\neq 0 \wedge m+pq+(n-q)(2p-1)+1\neq 0$, **then**

$$\int x^m\left(a x^q+b x^n+c x^{2n-q}\right)^p dx \to$$
$$\left(x^{m-n+q+1}\left(b(n-q)p+c(m+pq+(n-q)(2p-1)+1)x^{n-q}\right)\left(a x^q+b x^n+c x^{2n-q}\right)^p\right)/\left(c(m+p(2n-q)+1)(m+pq+(n-q)(2p-1)+1)\right)+$$
$$\frac{(n-q)p}{c(m+p(2n-q)+1)(m+pq+(n-q)(2p-1)+1)}\cdot$$
$$\int x^{m-(n-2q)}\left(-ab(m+pq-n+q+1)+\left(2ac(m+pq+(n-q)(2p-1)+1)-b^2(m+pq+(n-q)(p-1)+1)\right)x^{n-q}\right)\left(a x^q+b x^n+c x^{2n-q}\right)^{p-1}dx$$

Program code:

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  x^(m-n+q+1)*(b*(n-q)*p+c*(m+p*q+(n-q)*(2*p-1)+1)*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^p/(c*(m+p*(2*n-q)+1)*(m+p*q+(n-q)*(2*p-1)+1))
  (n-q)*p/(c*(m+p*(2*n-q)+1)*(m+p*q+(n-q)*(2*p-1)+1))*
    Int[x^(m-(n-2*q))*
      Simp[-a*b*(m+p*q-n+q+1)+(2*a*c*(m+p*q+(n-q)*(2*p-1)+1)-b^2*(m+p*q+(n-q)*(p-1)+1))*x^(n-q),x]*
      (a*x^q+b*x^n+c*x^(2*n-q))^(p-1),x] /;
FreeQ[{a,b,c},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GtQ[p,0] &&
  RationalQ[m,q] && GtQ[m+p*q+1,n-q] && NeQ[m+p*(2*n-q)+1,0] && NeQ[m+p*q+(n-q)*(2*p-1)+1,0]
```

3: $\int x^m\left(a x^q+b x^n+c x^{2n-q}\right)^p dx$ when $q<n \wedge p\notin\mathbb{Z} \wedge b^2-4ac\neq 0 \wedge n\in\mathbb{Z}^+ \wedge p>0 \wedge m+pq+1<-(n-q) \wedge m+pq+1\neq 0$

Derivation: Generalized trinomial recurrence 1a with $A=1$ and $B=0$

Rule: If $q<n \wedge p\notin\mathbb{Z} \wedge b^2-4ac\neq 0 \wedge n\in\mathbb{Z}^+ \wedge p>0 \wedge m+pq+1\leq-(n-q)+1 \wedge m+pq+1\neq 0$, **then**

$$\int x^m\left(a x^q+b x^n+c x^{2n-q}\right)^p dx \to \frac{x^{m+1}\left(a x^q+b x^n+c x^{2n-q}\right)^p}{m+pq+1}-\frac{(n-q)p}{m+pq+1}\int x^{m+n}\left(b+2c x^{n-q}\right)\left(a x^q+b x^n+c x^{2n-q}\right)^{p-1}dx$$

Program code:

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  x^(m+1)*(a*x^q+b*x^n+c*x^(2*n-q))^p/(m+p*q+1) -
  (n-q)*p/(m+p*q+1)*Int[x^(m+n)*(b+2*c*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p-1),x] /;
FreeQ[{a,b,c},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GtQ[p,0] &&
  RationalQ[m,q] && LeQ[m+p*q+1,-(n-q)+1] && NeQ[m+p*q+1,0]
```

4: $\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land p > 0 \land m + pq + 1 > -(n - q) \land m + p(2n - q) + 1 \neq 0$

Derivation: Generalized trinomial recurrence 1a with $A = 0$, $B = 1$ and $m = m - n$

Derivation: Generalized trinomial recurrence 1b with $A = 1$ and $B = 0$

Rule: If $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land p > 0 \land m + pq + 1 > -(n - q) \land m + p(2n - q) + 1 \neq 0$, then

$$\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \to \frac{x^{m+1} \left(a x^q + b x^n + c x^{2n-q}\right)^p}{m + p(2n - q) + 1} + \frac{(n - q) p}{m + p(2n - q) + 1} \int x^{m+q} \left(2a + b x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^{p-1} dx$$

Program code:

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  x^(m+1)*(a*x^q+b*x^n+c*x^(2*n-q))^p/(m+p*(2*n-q)+1) +
  (n-q)*p/(m+p*(2*n-q)+1)*Int[x^(m+q)*(2*a+b*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p-1),x] /;
FreeQ[{a,b,c},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GtQ[p,0] &&
  RationalQ[m,q] && GtQ[m+p*q+1,-(n-q)] && NeQ[m+p*(2*n-q)+1,0]
```

8. $\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land p < -1$

1: $\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land p < -1 \land m + pq + 1 == -(n - q)(2p + 3)$

Derivation: Generalized trinomial recurrence 2b with $A = 1$, $B = 0$ and $m + pq + 1 == -(n - q)(2p + 3)$

Rule: If $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land p < -1 \land m + pq + 1 == -(n - q)(2p + 3)$, then

$$\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \to$$

$$-\frac{x^{m-q+1} \left(b^2 - 2ac + bc x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^{p+1}}{a(n - q)(p + 1)\left(b^2 - 4ac\right)} + \frac{2ac - b^2(p + 2)}{a(p + 1)\left(b^2 - 4ac\right)} \int x^{m-q} \left(a x^q + b x^n + c x^{2n-q}\right)^{p+1} dx$$

Program code:

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  -x^(m-q+1)*(b^2-2*a*c+b*c*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/(a*(n-q)*(p+1)*(b^2-4*a*c)) +
  (2*a*c-b^2*(p+2))/(a*(p+1)*(b^2-4*a*c))*
    Int[x^(m-q)*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1),x] /;
FreeQ[{a,b,c},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1] &&
  RationalQ[m,p,q] && EqQ[m+p*q+1,-(n-q)*(2*p+3)]
```

2: $\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land p < -1 \land m + pq + 1 > 2(n-q)$

Derivation: Generalized trinomial recurrence 2a with $A = 0$, $B = 1$ and $m = m - n + q$

Rule: If $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land p < -1 \land m + pq + 1 > 2(n-q)$, then

$$\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \longrightarrow$$

$$-\frac{x^{m-2n+q+1}\left(2a + b x^{n-q}\right)\left(a x^q + b x^n + c x^{2n-q}\right)^{p+1}}{(n-q)(p+1)\left(b^2 - 4ac\right)} +$$

$$\frac{1}{(n-q)(p+1)\left(b^2 - 4ac\right)} \int x^{m-2n+q}\left(2a(m + pq - 2(n-q) + 1) + b(m + pq + (n-q)(2p+1) + 1)x^{n-q}\right)\left(a x^q + b x^n + c x^{2n-q}\right)^{p+1} dx$$

Program code:

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  -x^(m-2*n+q+1)*(2*a+b*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/((n-q)*(p+1)*(b^2-4*a*c)) +
  1/((n-q)*(p+1)*(b^2-4*a*c))*
    Int[x^(m-2*n+q)*(2*a*(m+p*q-2*(n-q)+1)+b*(m+p*q+(n-q)*(2*p+1)+1)*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1),x] /;
FreeQ[{a,b,c},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1] &&
  RationalQ[m,q] && GtQ[m+p*q+1,2*(n-q)]
```

3: $\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land p < -1 \land m + pq + 1 < n - q$

Derivation: Generalized trinomial recurrence 2b with $A = 1$ and $B = 0$

Rule: If $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land p < -1 \land m + pq + 1 < n - q$, then

$$\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \longrightarrow$$

$$-\frac{x^{m-q+1} \left(b^2 - 2ac + bc x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^{p+1}}{a (n-q) (p+1) \left(b^2 - 4ac\right)} + \frac{1}{a (n-q) (p+1) \left(b^2 - 4ac\right)} \cdot$$

$$\int x^{m-q} \left(b^2 (m + pq + (n-q)(p+1) + 1) - 2ac (m + pq + 2(n-q)(p+1) + 1) + bc (m + pq + (n-q)(2p+3) + 1) x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^{p+1} dx$$

Program code:

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  -x^(m-q+1)*(b^2-2*a*c+b*c*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/(a*(n-q)*(p+1)*(b^2-4*a*c)) +
  1/(a*(n-q)*(p+1)*(b^2-4*a*c))*
    Int[x^(m-q)*
      (b^2*(m+p*q+(n-q)*(p+1)+1)-2*a*c*(m+p*q+2*(n-q)*(p+1)+1)+b*c*(m+p*q+(n-q)*(2*p+3)+1)*x^(n-q))*
      (a*x^q+b*x^n+c*x^(2*n-q))^(p+1),x] /;
FreeQ[{a,b,c},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1] &&
  RationalQ[m,q] && LtQ[m+p*q+1,n-q]
```

4: $\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land p < -1 \land n - q < m + pq + 1 < 2(n - q)$

Derivation: Generalized trinomial recurrence 2a with $A = 1$ and $B = 0$

Derivation: Generalized trinomial recurrence 2b with $A = 0$, $B = 1$ and $m = m - n$

Rule: If $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land p < -1 \land n - q < m + pq + 1 < 2(n - q)$, then

$$\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \longrightarrow$$

$$\frac{x^{m-n+1} \left(b + 2 c x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^{p+1}}{(n-q)(p+1)\left(b^2 - 4ac\right)} -$$

$$\frac{1}{(n-q)(p+1)\left(b^2 - 4ac\right)} \int x^{m-n} \left(b(m + pq - n + q + 1) + 2c(m + pq + 2(n - q)(p + 1) + 1) x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^{p+1} dx$$

Program code:

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  x^(m-n+1)*(b+2*c*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/((n-q)*(p+1)*(b^2-4*a*c)) -
  1/((n-q)*(p+1)*(b^2-4*a*c))*
    Int[x^(m-n)*(b*(m+p*q-n+q+1)+2*c*(m+p*q+2*(n-q)*(p+1)+1)*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1),x] /;
FreeQ[{a,b,c},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1] &&
  RationalQ[m,q] && LtQ[n-q,m+p*q+1,2*(n-q)]
```

9. $\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land -1 \leq p < 0$

1: $\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land -1 \leq p < 0 \land m + pq + 1 == 2(n-q)$

Derivation: Generalized trinomial recurrence 3a with $A = 0$, $B = 1$ and $m = (-pq + 2(n-q) - 1) - n + q$

Rule: If $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land -1 \leq p < 0 \land m + pq + 1 == 2(n-q)$, then

$$\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \longrightarrow \frac{x^{m-2n+q+1} \left(a x^q + b x^n + c x^{2n-q}\right)^{p+1}}{2c(n-q)(p+1)} - \frac{b}{2c} \int x^{m-n+q} \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$$

Program code:

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  x^(m-2*n+q+1)*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/(2*c*(n-q)*(p+1)) -
  b/(2*c)*Int[x^(m-n+q)*(a*x^q+b*x^n+c*x^(2*n-q))^p,x] /;
FreeQ[{a,b,c},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GeQ[p,-1] && LtQ[p,0] &&
  RationalQ[m,q] && EqQ[m+p*q+1,2*(n-q)]
```

2: $\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land -1 \leq p < 0 \land m + pq + 1 == -2(n-q)(p+1)$

Derivation: Generalized trinomial recurrence 3b with $A = 1$, $B = 0$ and $m + pq + 1 == -2(n-q)(p+1)$

Rule: If $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land m + pq + 1 \neq 0 \land n \in \mathbb{Z}^+ \land -1 \leq p < 0 \land m + pq + 1 == -2(n-q)(p+1)$, then

$$\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \longrightarrow -\frac{x^{m-q+1} \left(a x^q + b x^n + c x^{2n-q}\right)^{p+1}}{2a(n-q)(p+1)} - \frac{b}{2a} \int x^{m+n-q} \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$$

Program code:

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  -x^(m-q+1)*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/(2*a*(n-q)*(p+1)) -
  b/(2*a)*Int[x^(m+n-q)*(a*x^q+b*x^n+c*x^(2*n-q))^p,x] /;
FreeQ[{a,b,c},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GeQ[p,-1] && LtQ[p,0] &&
  RationalQ[m,q] && EqQ[m+p*q+1,-2*(n-q)*(p+1)]
```

3: $\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land -1 \leq p < 0 \land m + p q + 1 > 2 (n - q)$

Derivation: Generalized trinomial recurrence 3a with $A = 0$, $B = 1$ and $m = m - n + q$

Note: If $-1 \leq p < 0$ and $m + p q + 1 > 2 (n - q)$, then $m + p q + 2 (n - q) p + 1 \neq 0$.

Rule: If $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land -1 \leq p < 0 \land m + p q + 1 > 2 (n - q)$, then

$$\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \longrightarrow$$

$$\frac{x^{m-2n+q+1} \left(a x^q + b x^n + c x^{2n-q}\right)^{p+1}}{c (m + p q + 2 (n - q) p + 1)} -$$

$$\frac{1}{c (m + p q + 2 (n - q) p + 1)} \int x^{m-2(n-q)} (a (m + p q - 2 (n - q) + 1) + b (m + p q + (n - q) (p - 1) + 1) x^{n-q}) \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$$

Program code:

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  x^(m-2*n+q+1)*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/(c*(m+p*q+2*(n-q)*p+1)) -
  1/(c*(m+p*q+2*(n-q)*p+1))*
    Int[x^(m-2*(n-q))*(a*(m+p*q-2*(n-q)+1)+b*(m+p*q+(n-q)*(p-1)+1)*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^p,x] /;
FreeQ[{a,b,c},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GeQ[p,-1] && LtQ[p,0] &&
  RationalQ[m,q] && GtQ[m+p*q+1,2*(n-q)]
```

4: $\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land -1 \leq p < 0 \land m + pq + 1 < 0$

Derivation: Generalized trinomial recurrence 3b with $A = 1$ and $B = 0$

Rule: If $q < n \land p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land -1 \leq p < 0 \land m + pq + 1 < 0$, then

$$\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \longrightarrow$$

$$\frac{x^{m-q+1} \left(a x^q + b x^n + c x^{2n-q}\right)^{p+1}}{a (m + pq + 1)} -$$

$$\frac{1}{a (m + pq + 1)} \int x^{m+n-q} \left(b (m + pq + (n-q)(p+1) + 1) + c (m + pq + 2 (n-q)(p+1) + 1) x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$$

Program code:

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  x^(m-q+1)*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/(a*(m+p*q+1)) -
  1/(a*(m+p*q+1))*
    Int[x^(m+n-q)*(b*(m+p*q+(n-q)*(p+1)+1)+c*(m+p*q+2*(n-q)*(p+1)+1)*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^p,x] /;
FreeQ[{a,b,c},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GeQ[p,-1] && LtQ[p,0] &&
  RationalQ[m,q] && LtQ[m+p*q+1,0]
```

10: $\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $p \notin \mathbb{Z}$

- **Derivation: Piecewise constant extraction**

- **Basis:** $\partial_x \frac{\left(a x^q + b x^n + c x^{2n-q}\right)^p}{x^{pq} \left(a + b x^{n-q} + c x^{2(n-q)}\right)^p} = 0$

- **Rule: If $p \notin \mathbb{Z}$, then**

$$\int x^m \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \longrightarrow \frac{\left(a x^q + b x^n + c x^{2n-q}\right)^p}{x^{pq} \left(a + b x^{n-q} + c x^{2(n-q)}\right)^p} \int x^{m+pq} \left(a + b x^{n-q} + c x^{2(n-q)}\right)^p dx$$

- **Program code:**

```
Int[x_^m_.*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  (a*x^q+b*x^n+c*x^(2*n-q))^p/(x^(p*q)*(a+b*x^(n-q)+c*x^(2*(n-q)))^p)*
    Int[x^(m+p*q)*(a+b*x^(n-q)+c*x^(2*(n-q)))^p,x] /;
FreeQ[{a,b,c,m,n,p,q},x] && EqQ[r,2*n-q] && Not[IntegerQ[p]] && PosQ[n-q]
```

S: $\int u^m \left(a\,u^q + b\,u^n + c\,u^{2\,n-q}\right)^p \,dx$ when $u == d + e\,x$

- **Derivation: Integration by substitution**
- **Rule: If $u == d + e\,x$, then**

$$\int u^m \left(a\,u^q + b\,u^n + c\,u^{2\,n-q}\right)^p \,dx \;\longrightarrow\; \frac{1}{e}\,\text{Subst}\left[\int x^m \left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^p \,dx,\; x,\; u\right]$$

- **Program code:**

```
Int[u_^m_.*(a_.*u_^q_.+b_.*u_^n_.+c_.*u_^r_.)^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[x^m*(a*x^q+b*x^n+c*x^(2*n-q))^p,x],x,u] /;
FreeQ[{a,b,c,m,n,p,q},x] && EqQ[r,2*n-q] && LinearQ[u,x] && NeQ[u,x]
```